



## Game High-Level Description

Matching pennies is a zero-sum game used in game theory. For this assignment, we implement this game on a Smart Contract using *Solidity*. In our version of the game, we have two players who want to bet against each other. Each player will randomly pick a number 0 or 1 (we can think of it as true/false or head/tails). If both players place the same bet, playerA will get the reward of 1 ETH; else, if both players choose different numbers playerB wins. This is a zero-sum game since the reward of the winner is the loss of the other player. The diagram below illustrates the aforementioned winner/loser scheme we have described.

		Player A Bet	
		0	1
Player B Bet	0	<i>Player A</i>	<i>Player B</i>
	1	<i>Player B</i>	<i>Player A</i>

Only two players can participate in the game, and they have a specific time limit to complete their match so other players can play the game later. Each player who wants to join the game has to bet 1 ETH and pay extra transaction gas fees.

It is essential to mention that when a player wants to play, there is a two-step process. Firstly, he has to obscure his bet. He needs to do this because any transaction data are visible on public- permissionless blockchains. To hide a bet, we use a program to hash its value using a random salt. This program can be found at the end of this report.

$$\text{Obscured Bet} = H(\text{bet}||\text{salt}), \quad \text{where } H : \text{Keccak256Hash}$$

Using this technique, when playerA submits his bet on the Smart Contract, nobody can view his real prediction and thus not cheat on him. Afterward, playerB has to place his bet in the same manner.

When this phase completes, players' bets are locked, and they can not make any changes. Subsequently, we move to the second phase of the game, where each player has to provide his real bet and salt to validate their initial inputs.

In addition, whenever a player interacts with the Smart Contract, an event is emitted to notify his opponent to progress. Each player has 10 minutes to react, else the one who played last can request a refund and cancel the game. The player who did not respond in time loses his ETH, which are kept by the contract.

Finally, any of the two players can call the Smart Contract evaluate function to calculate the winner. The winner can immediately withdraw his reward from the Smart Contract and reset the game settings allowing another pair of players to compete.

There follows a detailed description of our Smart Contract Variables, Functions, and Events:

## **Variables**

*\_playDeadline*: This is a public uint256 that stores the last time a player interacted with the game.

*\_playedLast*: This is a public uint256 that stores the address of the last player.

*\_adr\_playerA*: This is a public uint256 that stores the address of playerA

*\_adr\_playerB*: This is a public uint256 that stores the address of playerB

*\_bets*: This is a public mapping which stores a players address and its corresponding Bet. The bet is a struct as shown below

*Bet*: This is a struct consist of a string which is the real bed, a bytes32 which represents the hidden bet , and a boolean variable that indicates if the bet is valid.

*\_locked*: This is a public uint8 that takes values 0,1. When its value is 1 the Smart Contract is considered locked, and no one else can place new bets.

*\_playersJoined*: This is a public uint8 that stores how many players joined the game on a round. It takes values 0,1,2.

*\_winner*: This variable stores the address of the winner.

## **Events**

*event Play*: Whenever a player interacts with the Smart Contract to place his bet, a new event is emitted to notify the other player to place his bet. This event displays information such as the address of the player who interacted last and his index.

*event Winner*: Whenever a player wins the game, this event is emitted, so the winning entity knows. This event displays the winner address.

*event NewGame*: Whenever the Smart Contract is available and can host a new round this event is emitted.

## **Functions**

*giveHiddenBet()*: This is a payable function. The player provides a bytes32 value representing his bet. The function requirements are:

1. A free spot to play.
2. The player to send 1 ETH.

3. The player has not already placed another bet.

After these checks are complete, the Smart Contract variables are updated, and the player's bet is final. Eventually, this function emits an event to notify the other player to progress.

*giveRealBet()*: This external function can be called after both players have placed their hidden bet. It receives as input the real bet and salt of each player and validates and updates their hidden bet. Eventually, this function emits an event to notify the other player to progress.

*evaluateWinner()*: This external function is called to determine which player won the game. The constraint for this function is that both users' bets are valid. Eventually, this function emits an event to notify players about the game outcome.

*requestRefund()*: This is an external function and can be called by the player who played last, and only if 10 minutes have passed from his previous interaction. Then the *gameReset* function is called, and one ether is returned to him.

*withdraw()*: This is an external function that can be called only by the game-winner. If this requirement is satisfied, the *gameReset()* function is called, and two ether are returned to the winner.

*gameReset()*: This is an internal function that can be called by *requestRefund()* and *withdraw()* functions. Its purpose is to reset the Smart Contract variables and prepare the game to host a new pair of players. Finally, this function delivers the requested amount of ether to the proper recipient.

## Gas Costs Evaluation

This section measures and evaluates the gas cost we expect to have when deploying and interacting with the Smart Contract.

Gas costs appear in the following table, and they are unquestionably high. The Smart Contract owner has to pay 1,801,345 gas units to deploy the code, which is approximately \$995 (price on 26/10/21). This is definitely a high amount of gas which makes the Smart Contract not so gas efficient.

**Gas Costs**

<b><i>Contract Owner Fees</i></b>		
Contract Deployment	1,801,345	

  

<b><i>Players Fees</i></b>	<b><i>Player A</i></b>	<b><i>Player B</i></b>
<i>giveHiddenBet()</i>	135,643	84,620
<i>giveRealBet()</i>	81,007	83,807
<i>evaluate()</i>	40,423	
<i>withdraw()</i>	43,717	
<i>requestRefund()</i>	54,648	

In regards to the players interacting with the Smart Contract, the fees are much lower. Namely, during the initial phase of submitting a hidden bet, *playerA* has to pay 135,643 and *playerB*

84,620 gas units, translating to approximately \$61 and \$44, respectively. We can notice a considerable difference between the fees paying each player, making this a bit unfair.

For the next step, both players need to call the `giveRealBet()` function, which will cost them 81,007(\$42) and 83807 (\$43) accordingly. Finally, the evaluation, withdrawal, and refund functions will cost 40,423(\$20), 43,717(\$20), 54,648(\$28) gas units.

Reflecting on the above measurements, it is apparent that *playerA* has a disadvantage against *playerB* in terms of gas fairness. In total, *playerA* has to pay 216,650 gas units while *playerB* 168,427 gas units, occurring a difference of 48,223. To mitigate this imbalance, we propose that *playerB* call the `evaluateWinner()` function narrowing the gap to 7800 gas units. Regarding refund requests and withdrawals, it is more than fair that the transaction invoker pays the proportional fee for himself.

Potentially, we could argue that the Smart Contract can be improved in terms of gas efficiency. Such improvements would be to use a different structure to store the players' bets. Using a string in the `Bet` struct is not ideal, but the validation process could not be achieved using `bytes32` data type. There was an issue when it came to concatenating the real bet and salt using the Remix environment.

## Potential Hazards and Vulnerabilities

Developing a Smart Contract can always be challenging. This is because, on a public-permissionless blockchains, everything is observable by everyone. This fact makes it difficult when it comes to securing users' data. Besides that, a developer has to be alert to write code that is attack-resistant. Smart Contracts expose a broad spectrum of vulnerabilities, enabling an adversarial entity to exploit them for his interest.

When developing the Matching Pennies game, we took into consideration possible attacks. The following list presents vulnerabilities and mechanisms to moderate them.

***DoS(Denial of Service) - Griefing:*** An attacker attempts to make a Smart Contract get stuck when executed. In the case of our implementation, a player might grieve and stop playing to halt the Smart Contract or make his opponent lose money. This is not wanted since the other player's ETH will get stuck, and no one else will be able to play the game.

***Mitigation:*** To counter this type of attack, we implemented a time limit mechanism. When a player interacts with the Smart Contract, a timer is initiated. The other player has only 10 minutes to play his move. If the time limit expires, the last player can request a refund and cancel the game. The funds of the lapsed player will be kept as punishment. If a single player tries to halt the program (i.e., only one player joins the game and his opponent refuse), the Smart Contract owner can join, to force the first player to play. If the first player denies playing, he will lose his money since the contract owner can request a refund, which will reset the game. Using this technique, the Smart Contract can never halt.

**Re-Entrancy:** An attacker might try to take advantage of the Smart Contract withdraw function by executing a re-entrancy attack. In more detail, when he invokes a withdrawal transaction, he can drive his transaction to a malicious fallback function on another Smart Contract that can recursively call again and again the withdraw function, trying to get more ETH.

**Mitigation:** In order to avoid such unpleasant attacks, we execute the code of the Smart Contract in a particular way. When there is a withdrawal invocation, we check some constraints to ensure that the transaction sender is allowed to withdraw ETH. After that, we will pass any updates to the state of the Smart Contract and eventually make the call that sends the requested ETH to the recipient. It is important to make the ETH transfer after changing the Smart Contract state. If a reentrancy attack occurs on the next transaction, it will be stopped because function constraints will evaluate the transaction according to the updated state.

**Front-Running:** This attack happens on the Miner level. An attacker might clone your transaction and put a much higher gas limit on it. This results in the inclusion of his transaction to the next block instead of yours. This is inconvenient since another player might steal your spot in the game.

**Mitigation:** There is no straightforward solution since the problem lies at the transaction mining level. For the Matching Pennies game, front-running will not have a significant effect, except that a player might steal the position of another one. The disfavored player will have the opportunity to play in another moment.

**Overflow/Underflow:** Matching Pennies game does not face this problem since it does not receive any integer value from the transaction sender.

**Randomness Source Exposure:** Matching Pennies game does not face this problem since it does not use any random value during the code execution.

**Delegation:** Matching Pennies game does not face this problem since it does not use code from other Smart Contracts or Libraries.

### Other good practices:

- Use `call()` instead of `transfer()` or `send()`. Using `call()` might be insecure, but `transfer()` and `send()` can forward only 2300 gas. In the future gas costs might change, and 2300 gas might not be enough. As a result using `call()` properly is the right approach. Below we can see an example of using `call()` function and handling its outcome correctly.

```
1 (bool success, ) = msg.sender.call.value(amount)("");  
2 require(success, "Transfer failed.");
```

- If you need to have a callback function, keep it simple.
- Any public-permissionless blockchain reveals transaction details to the ledger. Due to this fact, in data-sensitive applications such as the Matching Pennies game, we need to hide users' data. We can do this by following a two-phase process: committing a value obscured by hashing the bet and some salt and then exposing the real value.

## Security vs Performance

Developing a secure Smart Contract while providing a great user experience is a complex task. We can not always have both. For example, we prefer to have push over pull design when it comes to withdrawing ether. This is because we want to avoid out-of-gas exceptions and isolate each money transfer to its own transaction. The downside of this approach is that the users have to take more steps.

Furthermore, the user needs to know how to cover any sensitive information before submitting a transaction to the blockchain. This implies that users have to be educated on blockchain systems. This is not always feasible. Also, trying to secure our code by having more constraints, we make our Smart Contract less gas efficient and expensive to use.

Besides that, when it comes to writing secure code, the programming language used might negatively affect the final product. This is because pretty new languages such as *Solidity* were designed having specific specs in mind. We can reinforce *Solidity's* core functionality by using libraries such as *OpenZeppelin*, but the solution will not always fit.

In conclusion, we can argue that when it comes to Smart Contract development, we have to make some crucial decisions between security, performance, and usability. Being able to find the golden mean in this problem can be very difficult.

## Fellow Student Contract Analysis

In this section, you can find the review of a Smart Contract, a fellow student implemented.

### Vulnerabilities :

1. Used *trasnfer()* function to transfer rewards or refunds.
2. Update the Smart Contract state after transferring funds.
3. A player is allowed to play the game by himself.
4. The *claimReward()* function can be called by both players.

### Exploitation :

1. The Smart Contract developer took the decision to use *trasnfer()* function instead of *call()*. From one point of view this implementation is logical if we consider that *trasnfer()* was introduced after the DAO attack to mitigate Re-Entrancy attacks. However, *trasnfer()* does not allow to forward more than 2300 gas. In the future, if gas costs change the Smart Contract might be problematic.
2. When someone tries to claim a reward or a refund, funds are transfered (lines 9 - 16 below) before the Smart Contract variables are updated (lines 18 - 22 below). In the example below an adversarial player1 might try to call *gameReset()* in line 1 before any bet is revealed. So, he will validly pass the *require* statements and procced to the line 11. From this point he will be able to perform a Re-Entrancy attack by calling again *gameReset()* and since the Smart Contract variables have not been updated yet, the attacker will pass any constraint and be able to withdraw 2 ETH instead of 1 that is allowed.

```

1 function resetGame () public { // allows people to play again, but doesn
  't deal with deposited ETH
2   require (msg.sender == player1 || msg.sender == player2, "You are
    not a player in this game" );
3
4   // if both players joined the game and one of them have revealed
    their choice, then the reset is not allowed
5   if (player1 != address(0) && player2 != address(0)) {
6     require ( !bets[player1].isValid && !bets[player2].isValid, "A
        player has already revealed choice, so you can no longer
        reset, game needs to be finished" );
7   }
8   // refund each player
9   if (player1 != address(0) ) {
10    require ( !bets[player1].isValid, "Player 1 has already revealed
        choice, so you can no longer reset, game needs to be
        finished" );
11    refundBet(payable(player1));
12  }
13  if ( player2 != address(0)) {
14    require ( !bets[player2].isValid, "Player 2 has already revealed
        choice, so you can no longer reset, game needs to be
        finished" );
15    refundBet(payable(player2));
16  }
17
18  choices[player1] = false; // these two lines allow each player to
    play again possibly
19  choices[player2] = false;
20
21  player1 = address(0); // allows another two players to play
22  player2 = address(0);
23 }

```

3. The particular Smart Contract enables a player to play the game by himself. As a result, an adversarial entity that wants to stuck the code forever, can place a bet with himself twice and leave. Consequently, the game will be locked and unavailable to other players.
4. The *claimReward()* function can be called by both players. As you can observe in the following figure, *claimReward()* can be called by both players at any time. According to lines 6 - 9 the actual winner is the one who calls the function first. As a result, the Smart Contract is vulnerable to Front-Running attacks. For instance, the loser of the game, might call the *claimReward()* function using more gas and steal the reward.

```

1 function claimReward() external {
2   require (msg.sender == player1 || msg.sender == player2, "You are
    not a player in this game" );
3   require ( bets[player1].isValid, "Player 1 did not reveal a valid
    bet" );
4   require ( bets[player2].isValid, "Player 2 did not reveal a valid
    bet" );
5
6   if (choices[player1] == choices[player2]) {
7     winner = payable(msg.sender);
8     transferReward(); // calls on a separate function to transfer
        the 2 Eth
9   }
10  else {

```

```

11         winner = payable(bets[msg.sender].opponent);
12         transferReward();
13     }
14
15     // reset the game for the next two players
16
17     choices[player1] = false;
18     choices[player2] = false;
19
20     player1 = address(0);
21     player2 = address(0);
22 }

```

## Smart Contract Execution History

Owner Address: 0x79BE6e946368520419BF4A20aC45b28fd3a5b2bA

Player A Address: 0xE083f2644739ef27EC126e42288253F0b9AdFB85

Player B Address: 0xd29Fd58d75aE640415D23166802EA3bd66Ddfd04

Player A Balance: 2 ETH

Player B Balance: 2 ETH

- **Phase 0 - Smart Contract deployment**

Deployment tx: 0x66014298bc638b7a72b29ef643c15960910470bed2918bcceaec0bfdeda76ae

Contract Address: 0xF87a42464eEf144fF0C81cE8d5E927548CF4695D

- **Phase 1 - Submit hidden value**

Player A input: 0x3eb0fa86b29ff88fdd4458cd1f554dd6ad43237a86e38c862ab6c440a387964

Player A tx: 0x742e2b4bd030a2e1130fae0f5f9ff413407fd0502af2a1106dec3d9461453e15

Player B input: 0xf7f905159c4867bf40ccc7667b940bf77402f6daddc3055b3b2256cb0a291365

Player B tx: 0xfe3bdff40b02b116befa96647f6b31bf8a80c212aeeda5b7b20774c6994b981

- **Phase 2 - Submit real values**

Player A input: 0, 123

Player A tx: 0xb497dd616183ecccc8eba62c2989e5d4661e4af39b28bf9b43b63b2114fc1742

Player B input: 0, 234

Player B tx: 0xba1edb21440455e86c75a59f74205cd6874c7c993f2a866e3ce198b8232ceb9e

- **Phase 3 - Calculating Winner**

Evaluation tx: 0x623271aba3304cb1a4983979ae5521a3edb0989dba845982e9ea38d36be9bdfa



- Phase 4 - Winner withdrawal

Withdrawal tx: 0x91686c19055f98eb6b2450e0804b537dc6f62ce3b2ad843fcf47b71a0f4cdf32

Player A Balance: 2.9997 ETH

Player B Balance: 0.9998 ETH

## Implementation Code

```
1  pragma solidity 0.8.0;
2
3  /// @title Matching pennies game
4  /// @author Erodotos Demetriou
5  contract Game {
6      uint256 public _playDeadline;
7      address public _playedLast;
8      address public _adr_playerA;
9      address public _adr_playerB;
10
11     mapping(address => Bet) public _bets;
12
13     struct Bet {
14         string _realBet;
15         bytes32 _hiddenBet;
16         bool _isValid;
17     }
18
19     uint8 public _locked = 0;
20     uint8 public _playersJoined = 0;
21     address public _winner;
22
23     event Play(address indexed _playerAddress, uint8 _playerNumber);
24     event WinnerAnnounced(address indexed _winner);
25     event NewGame(string _newGame);
26
27     /// @notice Takes 1 ETH as bet stake and set contract state accordingly
28     /// @param _bet This is an obscured 32-byte string produced after
29     /// hashing (real_bet || salt)
30     function giveHiddenBet(bytes32 _bet) public payable {
31         // Perform checks
32         require(
33             _locked == 0,
34             "There are already 2 players. Wait for the next game to start!"
35         );
36         require(msg.value == 1 ether, "You must bet 1 ETH");
37         require(
38             _bets[msg.sender]._hiddenBet == bytes32(0),
39             "You have already put your bet"
40         );
41
42         // Change the smart contract state
43         _playersJoined += 1;
44         _bets[msg.sender]._hiddenBet = _bet;
45         _playDeadline = block.timestamp + 10 minutes;
46         _playedLast = msg.sender;
```

```

47
48 // Lock the contract if both players beted
49 // and emit events to announce their participation
50 if (_playersJoined == 2) {
51     _locked = 1;
52     _adr_playerB = msg.sender;
53     emit Play(msg.sender, 2);
54 } else {
55     _adr_playerA = msg.sender;
56     emit Play(msg.sender, 1);
57 }
58 }
59
60 /// @notice Receives the players real bets
61 /// and their salt and check the initial bet validity
62 /// @param _realBet A string representing the real bet
63 /// @param _salt The salt that the message sender used
64 /// to create his initial obscured bet
65 function giveRealBet(string memory _realBet, string memory _salt)
    external {
66     require(_playersJoined == 2, "Wait for player #2 to join the game");
67     require(
68         keccak256(abi.encodePacked(_realBet, _salt)) ==
69         _bets[msg.sender]._hiddenBet,
70         "Error: Provided invalid input: Abort"
71     );
72
73     _bets[msg.sender]._realBet = _realBet;
74     _bets[msg.sender]._isValid = true;
75
76     _playedLast = msg.sender;
77     _playDeadline = block.timestamp + 10 minutes;
78 }
79
80 /// @notice Calculates the game winner
81 function evaluateWinner() external {
82     require(
83         _bets[_adr_playerA]._isValid && _bets[_adr_playerB]._isValid,
84         "Error: Players did not provide their real bet"
85     );
86
87     if (
88         keccak256(abi.encode(_bets[_adr_playerA]._realBet)) ==
89         keccak256(abi.encode(_bets[_adr_playerB]._realBet))
90     ) {
91         _winner = _adr_playerA;
92     } else if (
93         keccak256(abi.encode(_bets[_adr_playerA]._realBet)) !=
94         keccak256(abi.encode(_bets[_adr_playerB]._realBet))
95     ) {
96         _winner = _adr_playerB;
97     }
98
99     // Emit event
100    emit WinnerAnnounced(_winner);
101 }
102
103 /// @notice Let a player to stop the game and get
104 /// refund in case his opponent grieves
105 function requestRefund() external {

```

```

106     // Checks
107     require(
108         block.timestamp > _playDeadline &&
109         msg.sender == _playedLast &&
110         _winner == address(0),
111         "You are not allowed to request a refund yet!"
112     );
113
114     gameReset(1 ether);
115 }
116
117 /// @notice Allows the winner to withdraw his reward
118 function withdraw() external {
119     // Checks
120     require(msg.sender == _winner, "You are not the winner!");
121
122     gameReset(2 ether);
123 }
124
125 /// @notice Send money to the winner or the
126 /// refund requestor and reset game variables for a new round
127 function gameReset(uint256 _value) internal {
128     _locked = 0;
129     _winner = address(0);
130     _playersJoined = 0;
131     _bets[_adr_playerA] = Bet("", bytes32(0), false);
132     _bets[_adr_playerB] = Bet("", bytes32(0), false);
133     _adr_playerA = address(0);
134     _adr_playerB = address(0);
135     _playDeadline = 0;
136
137     // Reward/Refund transfer
138     (bool success, ) = msg.sender.call{value: _value}("");
139     require(success, "Error: Withdraw unsuccessful");
140
141     // Emmit event
142     emit NewGame("New game spots available");
143 }
144 }

```

```

1 // Program to generate obscured bet
2
3 package main
4 import (
5     "flag"
6     "fmt"
7
8     "github.com/ethereum/go-ethereum/crypto"
9 )
10 func main() {
11     bet := flag.String("bet", "0", "Provide bet option, 0 or 1!")
12     salt := flag.String("salt", "random", "Provide a random salt")
13     flag.Parse()
14
15     input := []byte(*bet + *salt)
16     hash := crypto.Keccak256Hash(input)
17     fmt.Println(hash)
18 }

```